# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=onetwopunch
pkgver=v1.0.0.r2.gd4ab4e8
pkgrel=1
pkgdesc='Use unicornscan to quickly scan all open ports, and then pass the open ports to nmap for detailed scans.'
groups=('blackarch' 'blackarch-scanner')
arch=('any')
url='https://github.com/superkojiman/onetwopunch'
license=('MIT')
depends=('nmap' 'unicornscan')
makedepends=('git')
source=("git+https://github.com/superkojiman/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  git describe --tags --long | sed 's/\([^-]*-g\)/r\1/;s/-/./g'
}

package() {
  cd $pkgname

  install -Dm 755 "$pkgname.sh" "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md
  install -Dm 644 LICENSE.md "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

